DYNARE SUMMER SCHOOL

Introduction to Dynare and local approximation.

Michel Juillard

June 11, 2018

Summer School website

http://www.dynare.org/summerschool/2018

DYNARE

1. computes the solution of deterministic models (arbitrary accuracy),
2. computes first, second and third order approximation to solution of stochastic models,
3. estimates (maximum likelihood or Bayesian approach) parameters of DSGE models, for linear and non-linear models.
4. check for identification of estimated parameters
5. computes optimal policy,
6. performs global sensitivity analysis of a model,
7. estimates BVAR and Markov-Switching Bayesian VAR models.
8. Macro language and reporting facility

DSGE models

- Structural models that use theory to solve identifiaction problems.
- Microeconomic foundations $\Rightarrow$ *nonlinear models*
- Intertemporal optimization $\Rightarrow$ *expectations matter*. Rational expectations.
- Stochastic shocks push the economic system away from equilibrium. Endogenous dynamics bring it back towards equilibrium.
- Mathematical difficulty: solving nonlinear stochastic forward-looking model under rational expectations.

The general problem

Deterministic, perfect foresight, case:

$$f(y_{t+1}, y_t, y_{t-1}, u_t) = 0$$

Stochastic case:

$$E_t \{f(y_{t+1}, y_t, y_{t-1}, u_t)\} = 0$$

y : vector of endogenous variables

u : vector of exogenous shocks

Solution methods

- For a deterministic, perfect foresight, it is possible to compute numerical trajectories for the endogenous variables
- In a a stochastic framework, the unknown is the decision function:

$$y_t = g(y_{t-1}, u_t)$$

For a large class of DSGE models, DYNARE computes approximated decision rules and transition equations by a perturbation method.

Computation of first order approximation

- Perturbation approach: recovering a Taylor expansion of the solution function from a Taylor expansion of the original model.
- A first order approximation is nothing else than a standard solution thru linearization.
- A first order approximation in terms of the logarithm of the variables provides standard log-linearization.

General model

$$E_t\left\{f(y_{t+1}, y_t, y_{t-1}, u_t)\right\} = 0$$

$$\begin{aligned} E(u_t) &= 0 \\ E(u_t u_t') &= \Sigma_u \\ E(u_t u_\tau') &= 0 \quad t \neq \tau \end{aligned}$$

y : vector of endogenous variables

u : vector of exogenous stochastic shocks

Timing assumptions

$$E_t \{f(y_{t+1}, y_t, y_{t-1}, u_t)\} = 0$$

- shocks u_t are observed at the beginning of period t,
- decisions affecting the current value of the variables y_t, are function of
  - the previous state of the system, y_{t-1},
  - the shocks u_t.

The stochastic scale variable

$$E_t \left\{ f(y_{t+1}, y_t, y_{t-1}, u_t) \right\} = 0$$

- At period t, the only unknown stochastic variable is y_{t+1}, and, implicitly, u_{t+1}.
- We introduce the *stochastic scale variable*, σ and the auxiliary random variable, ϵ_t, such that

$$u_{t+1} = \sigma \epsilon_{t+1}$$

The stochastic scale variable (continued)

$$E(\epsilon_t) = 0 \tag{1}$$

$$E(\epsilon_t \epsilon_t') = \Sigma_\epsilon \tag{2}$$

$$E(\epsilon_t \epsilon_\tau') = 0 \quad t \neq \tau \tag{3}$$

and

$$\Sigma_u = \sigma^2 \Sigma_\epsilon$$

Remarks

$$E_t \left\{ f(y_{t+1}, y_t, y_{t-1}, u_t) \right\} = 0$$

- The exogenous shocks may appear only at the current period (in the presentation, not in Dynare)
- There is no deterministic exogenous variables
- Not all variables are necessarily present with a lead and a lag
- Generalization to leads and lags on more than one period (nonlinear models require special care for lead terms)

Solution function

$$y_t = g(y_{t-1}, u_t, \sigma)$$

where σ is the stochastic scale of the model. If $\sigma = 0$, the model is deterministic. For $\sigma > 0$, the model is stochastic. Under some conditions, the existence of $g()$ function is proven via an implicit function theorem. See H. Jin and K. Judd "Perturbation methods for general dynamic stochastic models"

(http://web.stanford.edu/~judd/papers/PerturbationMethodRatEx.pdf)

Solution function (continued)

Then,

$$\begin{aligned} y_{t+1} &= g(y_t, u_{t+1}, \sigma) \\ &= g(g(y_{t-1}, u_t, \sigma), u_{t+1}, \sigma) \\ F(y_{t-1}, u_t, \epsilon_{t+1}, \sigma) & \\ &= f(g(g(y_{t-1}, u_t, \sigma), \sigma\epsilon_{t+1}, \sigma), g(y_{t-1}, u_t, \sigma), y_{t-1}, u_t) \end{aligned}$$

$$E_t \left\{ F(y_{t-1}, u_t, \epsilon_{t+1}, \sigma) \right\} = 0$$

The perturbation approach

- Obtain a Taylor expansion of the unkown solution function in the neighborhood of a problem that we know how to solve.
- The problem that we know how to solve is the deterministic steady state.
- One obtains the Taylor expansion of the solution for the Taylor expansion of the original problem.
- One consider two different perturbations:
  1. points in the neighborhood from the steady sate,
  2. from a deterministic model towards a stochastic one (by increasing σ from a zero value).

The perturbation approach (continued)

- The Taylor approximation is taken with respect to y_{t-1}, u_t and σ, the arguments of the solution function

$$y_t = g(y_{t-1}, u_t, \sigma).$$

- At the deterministic steady state, all derivatives are deterministic as well.

Steady state

A deterministic steady state, $\bar{y}$, for the model satisfies

$$f(\bar{y}, \bar{y}, \bar{y}, 0) = 0$$

A model can have several steady states, but only one of them will be used for approximation.
Furthermore,

$$\bar{y} = g(\bar{y}, 0, 0)$$

First order approximation

Around $\bar{y}$:

$$
\begin{aligned}
E_t\left\{F^{(1)}(y_{t-1}, u_t, \epsilon_{t+1}, \sigma)\right\} &= \\
& E_t\Big\{f(\bar{y}, \bar{y}, \bar{y}, 0) + f_{y_+}\left(g_y\left(g_y \hat{y} + g_u u + g_\sigma \sigma\right) + g_u \sigma \epsilon' + g_\sigma \sigma\right) \\
& +f_{y_0}\left(g_y \hat{y} + g_u u + g_\sigma \sigma\right) + f_{y_-}\hat{y} + f_u u\Big\} \\
&= \quad 0
\end{aligned}
$$

with $\hat{y} = y_{t-1} - \bar{y}$, $u = u_t$, $\epsilon' = \epsilon_{t+1}$, $f_{y_+} = \frac{\partial f}{\partial y_{t+1}}$, $f_{y_0} = \frac{\partial f}{\partial y_t}$, $f_{y_-} = \frac{\partial f}{\partial y_{t-1}}$, $f_u = \frac{\partial f}{\partial u_t}$, $g_y = \frac{\partial g}{\partial y_{t-1}}$, $g_u = \frac{\partial g}{\partial u_t}$, $g_\sigma = \frac{\partial g}{\partial \sigma}$.

Certainty equivalence

$$
\begin{aligned}
& E_t\Big\{ f(\bar{y},\bar{y},\bar{y},0) + f_{y_+}\left(g_y\left(g_y\hat{y} + g_u u + g_\sigma\sigma\right) + g_u\sigma\epsilon' + g_\sigma\sigma\right) \\
& \quad + f_{y_0}\left(g_y\hat{y} + g_u u + g_\sigma\sigma\right) + f_{y_-}\hat{y} + f_u u\Big\} \\
= \; & f(\bar{y},\bar{y},\bar{y},0) + f_{y_+}\left(g_y\left(g_y\hat{y} + g_u u + g_\sigma\sigma\right) + g_u\sigma E_t\epsilon' + g_\sigma\sigma\right) \\
& + f_{y_0}\left(g_y\hat{y} + g_u u + g_\sigma\sigma\right) + f_{y_-}\hat{y} + f_u u \\
= \; & f(\bar{y},\bar{y},\bar{y},0) + f_{y_+}\left(g_y\left(g_y\hat{y} + g_u u + g_\sigma\sigma\right) + g_u\sigma + g_\sigma\sigma\right) \\
& + f_{y_0}\left(g_y\hat{y} + g_u u + g_\sigma\sigma\right) + f_{y_-}\hat{y} + f_u u \\
= \; & 0
\end{aligned}
$$

Taking the expectation

$$
\begin{aligned}
E_t\left\{F^{(1)}(y_{t-1}, u_t, \epsilon_{t+1}, \sigma)\right\} = \\
& f(\bar{y}, \bar{y}, \bar{y}, 0) + f_{y_+}\left(g_y\left(g_y\hat{y} + g_u u + g_\sigma\sigma\right) + g_\sigma\sigma\right) \\
& \left. + f_{y_0}\left(g_y\hat{y} + g_u u + g_\sigma\sigma\right) + f_{y_-}\hat{y} + f_u u\right\} \\
= \quad & \left(f_{y_+}g_y g_y + f_{y_0}g_y + f_{y_-}\right)\hat{y} + \left(f_{y_+}g_y g_u + f_{y_0}g_u + f_u\right)u \\
& + \left(f_{y_+}\left(g_y g_\sigma + g_\sigma\right) + f_{y_0}g_\sigma\right)\sigma \\
= \quad & 0
\end{aligned}
$$

Recovering g_y

$$\left(f_{y_+} g_y g_y + f_{y_0} g_y + f_{y_-}\right) \hat{y} = 0$$

Structural state space representation:

$$\begin{bmatrix} 0 & f_{y_+} \\ I & 0 \end{bmatrix} \begin{bmatrix} I \\ g_y \end{bmatrix} g_y \hat{y} = \begin{bmatrix} -f_{y_-} & -f_{y_0} \\ 0 & I \end{bmatrix} \begin{bmatrix} I \\ g_y \end{bmatrix} \hat{y}$$

or

$$\begin{bmatrix} 0 & f_{y_+} \\ I & 0 \end{bmatrix} \begin{bmatrix} y_t - \bar{y} \\ y_{t+1} - \bar{y} \end{bmatrix} = \begin{bmatrix} -f_{y_-} & -f_{y_0} \\ 0 & I \end{bmatrix} \begin{bmatrix} y_{t-1} - \bar{y} \\ y_t - \bar{y} \end{bmatrix}$$

Structural state space representation

$$Dx_{t+1} = Ex_t$$

with

$$x_{t+1} = \left[\begin{array}{c} y_t - \bar{y} \\ y_{t+1} - \bar{y} \end{array} \right] \qquad x_t = \left[\begin{array}{c} y_{t-1} - \bar{y} \\ y_t - \bar{y} \end{array} \right]$$

- There are multiple solutions but we want a unique stable one.
- Need to discuss eigenvalues of this linear system.
- Problem when D is singular.

Real generalized Schur decomposition

Taking the real generalized Schur decomposition of the pencil $< E, D >$:

$$\begin{aligned} D &= QTZ \\ E &= QSZ \end{aligned}$$

with T, upper triangular, S quasi-upper triangular, $Q'Q = I$ and $Z'Z = I$.

Generalized eigenvalues

λ_i solves

$$\lambda_i D x_i = E x_i$$

For diagonal blocks on S of dimension 1 x 1:

- $T_{ii} \neq 0$: $\lambda_i = \frac{S_{ii}}{T_{ii}}$
- $T_{ii} = 0$, $S_{ii} > 0$: $\lambda_i = +\infty$
- $T_{ii} = 0$, $S_{ii} < 0$: $\lambda_i = -\infty$
- $T_{ii} = 0$, $S_{ii} = 0$: $\lambda_i \in \mathcal{C}$

A pair of complex eigenvalues

When a diagonal block of matrix S is a 2x2 matrix of the form
$\begin{bmatrix} S_{ii} & S_{i,i+1} \\ S_{i+1,i} & S_{i+1,i+1} \end{bmatrix}$,

- the corresponding block of matrix T is a diagonal matrix,
- $\left(S_{i,i}T_{i+1,i+1} + S_{i+1,i+1}T_{i,i}\right)^2 < -4S_{i+1,i}S_{i+1,i}T_{i,i}T_{i+1,i+1}$,
- there is a pair of conjugate eigenvalues

$$\lambda_i, \lambda_{i+1} =$$
$$\frac{S_{ii}T_{i+1,i+1} + S_{i+1,i+1}T_{i,i} \pm \sqrt{(S_{i,i}T_{i+1,i+1} - S_{i+1,i+1}T_{i,i})^2 + 4S_{i+1,i}S_{i+1,i}T_{i,i}T_{i+1,i+1}}}{2T_{i,i}T_{i+1,i+1}}$$

Applying the decomposition

$$
\begin{aligned}
D \begin{bmatrix} I \\ g_y \end{bmatrix} g_y \hat{y} &= E \begin{bmatrix} I \\ g_y \end{bmatrix} \hat{y} \\
\begin{bmatrix} T_{11} & T_{12} \\ 0 & T_{22} \end{bmatrix} \begin{bmatrix} Z_{11} & Z_{12} \\ Z_{21} & Z_{22} \end{bmatrix} \begin{bmatrix} I \\ g_y \end{bmatrix} g_y \hat{y} & \\
&= \begin{bmatrix} S_{11} & S_{12} \\ 0 & S_{22} \end{bmatrix} \begin{bmatrix} Z_{11} & Z_{12} \\ Z_{21} & Z_{22} \end{bmatrix} \begin{bmatrix} I \\ g_y \end{bmatrix} \hat{y}
\end{aligned}
$$

Selecting the stable trajectory

To exclude explosive trajectories, one imposes

$$Z_{21} + Z_{22} g_y = 0$$

$$g_y = -Z_{22}^{-1} Z_{21}$$

A unique stable trajectory exists if Z_{22} is non-singular: there are as many roots larger than one in modulus as there are forward–looking variables in the model (Blanchard and Kahn condition) and the rank condition is satisfied.

An alternative algorithm: Cyclic reduction

- Solving
$$A_0 + A_1 X + A_2 X^2$$
- Iterate
$$\begin{aligned}
A_0^{(k+1)} &= -A_0^{(k)}(A_1^{(k)})^{-1}A_0^{(k)},\\
A_1^{(k+1)} &= A_1^{(k)} - A_0^{(k)}(A_1^{(k)})^{-1}A_2^{(k)} - A_2^{(k)}(A_1^{(k)})^{-1}A_0^{(k)},\\
A_2^{(k+1)} &= -A_2^{(k)}(A_1^{(k)})^{-1}A_2^{(k)},\\
\widehat{A}_1^{(k+1)} &= \widehat{A}_1^{(k)} - A_2^{(k)}(A_1^{(k)})^{-1}A_0^{(k)}.
\end{aligned}$$
  for $k = 1, \ldots$ with $A_0^{(1)} = A_0$, $A_1^{(1)} = A_1$, $A_2^{(1)} = A_2$, $\widehat{A}_1^{(1)} = A_1$ and until $||A_0^{(k)}||_\infty < \epsilon$ and $||A_2^{(k)}||_\infty < \epsilon$.
- Then
$$X \approx -(\widehat{A}_1^{(k+1)})^{-1}A_0$$

Recovering g_u

$$f_{y_+} g_y g_u + f_{y_0} g_u + f_u = 0$$

$$g_u = -\left(f_{y_+} g_y + f_{y_0}\right)^{-1} f_u$$

Recovering g_σ

$$f_{y_+} g_y g_\sigma + f_{y_0} g_\sigma = 0$$

$$g_\sigma = 0$$

Yet another manifestation of the certainty equivalence property of first order approximation.

First order approximated decision function

$$y_t = \bar{y} + g_y \hat{y} + g_u u$$

$$\begin{aligned} E\{y_t\} &= \bar{y} \\ \Sigma_y &= g_y \Sigma_y g_y' + \sigma^2 g_u \Sigma_\epsilon g_u' \end{aligned}$$

The variance is solved for with an algorithm for discrete time Lyapunov equations.

A simple RBC model

Consider the following model of an economy.

- Representative agent preferences

$$U = \sum_{t=1}^{\infty} \left(\frac{1}{1+\rho} \right)^{t-1} E_t \left[\log\left(C_t\right) - \frac{L_t^{1+\gamma}}{1+\gamma} \right].$$

  The household supplies labor and rents capital to the corporate sector.
  - L_t is labor services
  - $\rho \in (0, \infty)$ is the rate of time preference
  - $\gamma \in (0, \infty)$ is a labor supply parameter.
  - C_t is consumption,
  - w_t is the real wage,
  - r_t is the real rental rate

RBC Model (continued)

- The household faces the sequence of budget constraints

$$K_t = K_{t-1}(1-\delta) + w_t L_t + r_t K_{t-1} - C_t,$$

  where
  - K_t is capital at the end of period
  - $\delta \in (0,1)$ is the rate of depreciation
- The production function is given by the expression

$$Y_t = A_t K_{t-1}^{\alpha} \left((1+g)^t L_t \right)^{1-\alpha}$$

  where $g \in (0,\infty)$ is the growth rate and α and β are parameters.
- A_t is a technology shock that follows the process

$$A_t = A_{t-1}^{\lambda} \exp(e_t),$$

  where e_t is an i.i.d. zero mean normally distributed error with standard deviation σ_1 and $\lambda \in (0,1)$ is a parameter.

The household problem

Lagrangian

$$L = \max_{C_t, L_t, K_t} \sum_{t=1}^{\infty} \left(\frac{1}{1+\rho} \right)^{t-1} E_t \Bigg[\log\left(C_t\right) - \frac{L_t^{1+\gamma}}{1+\gamma}$$

$$- \mu_t \left(K_t - K_{t-1}\left(1-\delta\right) - w_t L_t - r_t K_{t-1} + C_t\right) \Bigg]$$

First order conditions

$$\frac{\partial L}{\partial C_t} = \left(\frac{1}{1+\rho} \right)^{t-1} \left(\frac{1}{C_t} - \mu_t \right) = 0$$

$$\frac{\partial L}{\partial L_t} = \left(\frac{1}{1+\rho} \right)^{t-1} \left(L_t^{\gamma} - \mu_t w_t \right) = 0$$

$$\frac{\partial L}{\partial K_t} = -\left(\frac{1}{1+\rho} \right)^{t-1} \mu_t + \left(\frac{1}{1+\rho} \right)^{t} E_t\left(\mu_{t+1}(1-\delta+r_{t+1})\right) = 0$$

First order conditions

Eliminating the Lagrange multiplier, one obtains

$$
\begin{aligned}
L_t^{\gamma} &= \frac{w_t}{C_t} \\
\frac{1}{C_t} &= \frac{1}{1+\rho} E_t \left(\frac{1}{C_{t+1}} (r_{t+1} + 1 - \delta) \right)
\end{aligned}
$$

The firm problem

$$\max_{L_t, K_{t-1}} A_t K_{t-1}^{\alpha} \left((1+g)^t L_t \right)^{1-\alpha} - r_t K_{t-1} - w_t L_t$$

First order conditions:

$$r_t = \alpha A_t K_{t-1}^{\alpha-1} \left((1+g)^t L_t \right)^{1-\alpha}$$

$$w_t = (1-\alpha) A_t K_{t-1}^{\alpha} \left((1+g)^t \right)^{1-\alpha} L_t^{-\alpha}$$

Goods market equilibrium

$$K_t + C_t = K_{t-1}(1-\delta) + A_t K_{t-1}^{\alpha}\left((1+g)^t L_t\right)^{1-\alpha}$$

Dynamic Equilibrium

$$
\begin{aligned}
\frac{1}{C_t} &= \frac{1}{1+\rho} E_t \left(\frac{1}{C_{t+1}} (r_{t+1} + 1 - \delta) \right) \\
L_t^{\gamma} &= \frac{w_t}{C_t} \\
r_t &= \alpha A_t K_{t-1}^{\alpha-1} \left((1+g)^t L_t \right)^{1-\alpha} \\
w_t &= (1-\alpha) A_t K_{t-1}^{\alpha} \left((1+g)^t \right)^{1-\alpha} L_t^{-\alpha} \\
K_t + C_t &= K_{t-1}(1-\delta) + A_t K_{t-1}^{\alpha} \left((1+g)^t L_t \right)^{1-\alpha}
\end{aligned}
$$

Existence of a balanced growth path

There must exist a growth rates g_c and g_k so that

$$(1+g_k)^t K_1 + (1+g_c)^t C_1 = \\ \frac{(1+g_k)^t}{1+g_K} K_0(1-\delta) + A\left(\frac{(1+g_k)^t}{1+g_k} K_0\right)^{\alpha} \left((1+g)^t L_t\right)^{1-\alpha}$$

So,

$$g_c = g_k = g$$

Stationarized model

Let's define

$$\widehat{C}_t = C_t/(1+g)^t$$
$$\widehat{K}_t = K_t/(1+g)^t$$
$$\widehat{w}_t = w_t/(1+g)^t$$

Stationarized model (continued)

$$\frac{1}{\widehat{C}_t(1+g)^t} = \frac{1}{1+\rho}E_t\left(\frac{1}{\widehat{C}_{t+1}(1+g)(1+g)^t}(r_{t+1}+1-\delta)\right)$$

$$L_t^{\gamma} = \frac{\widehat{w}_t(1+g)^t}{\widehat{C}_t(1+g)^t}$$

$$r_t = \alpha A_t\left(\widehat{K}_{t-1}\frac{(1+g)^t}{1+g}\right)^{\alpha-1}\left((1+g)^t L_t\right)^{1-\alpha}$$

$$\widehat{w}_t(1+g)^t = (1-\alpha)A_t\left(\widehat{K}_{t-1}\frac{(1+g)^t}{1+g}\right)^{\alpha}\left((1+g)^t\right)^{1-\alpha}L_t^{-\alpha}$$

$$\left(\widehat{K}_t+\widehat{C}_t\right)(1+g)^t = \widehat{K}_{t-1}\frac{(1+g)^t}{1+g}(1-\delta) + A_t\left(\widehat{K}_{t-1}\frac{(1+g)^t}{1+g}\right)^{\alpha}\left((1+g)^t L_t\right)^{1-\alpha}$$

Stationarized model (continued)

$$\frac{1}{\widehat{C}_t} = \frac{1}{1+\rho} E_t \left(\frac{1}{\widehat{C}_{t+1}(1+g)} (r_{t+1} + 1 - \delta) \right)$$

$$L_t^{\gamma} = \frac{\widehat{w}_t}{\widehat{C}_t}$$

$$r_t = \alpha A_t \left(\frac{\widehat{K}_{t-1}}{1+g} \right)^{\alpha-1} L_t^{1-\alpha}$$

$$\widehat{w}_t = (1-\alpha) A_t \left(\frac{\widehat{K}_{t-1}}{1+g} \right)^{\alpha} L_t^{-\alpha}$$

$$\widehat{K}_t + \widehat{C}_t = \frac{\widehat{K}_{t-1}}{1+g}(1-\delta) + A_t \left(\frac{\widehat{K}_{t-1}}{1+g} \right)^{\alpha} L_t^{1-\alpha}$$

Dynare implementation

```
var C K L w r A;
varexo e;

parameters rho delta gamma alpha lambda g;

alpha = 0.33;
delta = 0.1;
rho = 0.03;
lambda = 0.97;
gamma = 0;
g = 0.015;
```

Dynare implementation (continued)

```
model;
1/C=1/(1+rho)*(1/(C(+1)*(1+g)))*(r(+1)+1-delta);
L^gamma = w/C;
r = alpha*A*(K(-1)/(1+g))^(alpha-1)*L^(1-alpha);
w = (1-alpha)*A*(K(-1)/(1+g))^alpha*L^(-alpha);
K+C = (K(-1)/(1+g))*(1-delta)
     +A*(K(-1)/(1+g))^alpha*L^(1-alpha);
log(A) = lambda*log(A(-1))+e;
end;
```

Dynare implementation (continued)

```
steady_state_model;
A = 1;
r = (1+g)*(1+rho)+delta-1;
L = ((1-alpha)/(r/alpha-delta-g))*r/alpha;
K = (1+g)*(r/alpha)^(1/(alpha-1))*L;
C = (1-delta)*K/(1+g)
     +(K/(1+g))^alpha*L^(1-alpha)-K;
w = C;
end;

steady;
```

Dynare implementation (continued)

```
shocks;
var e; stderr 0.01;
end;

check;

stoch_simul(order=1);
```

Decision and transition functions

Dynare output:

POLICY AND TRANSITION FUNCTIONS

	C	K	L	w	r	A
Constant	1.003043	3.125296	0.906526	1.003043	0.145450	1.000000
K(-1)	0.144433	0.779746	-0.105500	0.144433	-0.042523	0
A(-1)	0.757723	1.149948	0.589451	0.757723	0.204452	0.970000
e	0.781158	1.185514	0.607681	0.781158	0.210776	1.000000

$$C_t = 1.003 + 0.144\left(K_{t-1} - \bar{K}\right) + 0.758\left(A_{t-1} - \bar{A}\right) + 0.781 e_t$$

Dating variables in Dynare

Dynare will automatically recognize predetermined and non–predetermined variables, but you must observe a few rules:

- period t variables are set during period t on the basis of the state of the system at period $t-1$ and shocks observed at the beginning of period t.
- therefore, stock variables must be on an end–of–period basis: investment of period t determines the capital stock at the end of period t.

Log–linearization

- Taking a log–linear approximation of a model is equivalent to take a linear approximation of a model with respect to the logarithm of the variables.
- In practice, it is sufficient to replace all occurences of variable X with $exp(LX)$ where $LX = \log X$.
- It is possible to make the substitution for some variables and not anothers. You wouldn't want to take a log approximation of a variable whose steady state value is negative ...
- There is no evidence that log–linearization is more accurate than simple linearization. In a growth model, it is often more natural to do a log–linearization.

The role of the Dynare preprocessor

- the Dynare toolbox solves generic problems
- the parser reads your *.mod file and translates it in specific Matlab files
- *filename*.m: main Matlab script for your model
- *filename*_static.m: static model
- *filename*_dynamic.m: dynamic model
- *filename*_steadystate2.m: steady state function
- *filename*_set_auxiliary_variables.m: auxiliary variables function

Second and third order approximation of the model

- Second and third order approximation of the solution function are obtained from second, respectively third, order approximation of the model.
- It requires only the solution of (tricky) linear problems.
- The stochastic scale of the model, σ, appears in the solution and breaks certainty equivalence

Second and third order decision functions

- Second order

$$
\begin{aligned}
y_t = \bar{y} + 0.5 g_{\sigma\sigma}\sigma^2 + g_y \hat{y} + g_u u \\
+ 0.5\left(g_{yy}(\hat{y} \otimes \hat{y}) + g_{uu}(u \otimes u)\right) + g_{yu}(\hat{y} \otimes u)
\end{aligned}
$$

- Third order

$$
\begin{aligned}
y_t = \bar{y} + \frac{1}{2} g_{\sigma\sigma}\sigma^2 + \frac{1}{6} g_{\sigma\sigma\sigma}\sigma^3 + \frac{1}{2} g_{y\sigma\sigma}\hat{y}\sigma^2 + \frac{1}{2} g_{u\sigma\sigma} u\sigma^2 \\
+ g_y \hat{y} + g_u u + \frac{1}{2}\left(g_{yy}(\hat{y} \otimes \hat{y}) + g_{uu}(u \otimes u)\right) \\
+ g_{yu}(\hat{y} \otimes u) + \frac{1}{6}\left(g_{yyy}(\hat{y} \otimes \hat{y} \otimes \hat{y}) + g_{uuu}(u \otimes u \otimes u)\right) \\
+ \frac{1}{2}\left(g_{yyu}(\hat{y} \otimes \hat{y} \otimes u) + g_{yuu}(\hat{y} \otimes \hat{y} \otimes u)\right)
\end{aligned}
$$

We can fix $\sigma = 1$.

Second order accurate moments

$$
\begin{aligned}
\Sigma_y &= g_y \Sigma_y g_y' + \sigma^2 g_u \Sigma_\epsilon g_u' \\
E\{y_t\} &= \bar{y} + (I - g_y)^{-1} \left(0.5 \left(g_{\sigma\sigma} + g_{yy} \vec{\Sigma}_y + g_{uu} \vec{\Sigma}_\epsilon\right)\right)
\end{aligned}
$$

Further issues

- Impulse response functions depend of state at time of shocks and history of future shocks.
- For large shocks second order approximation simulation may explode
  - pruning algorithm (Sims)
  - truncate normal distribution (Judd)

An asset pricing model

Urban Jermann (1998) “Asset pricing in production economies” *Journal of Monetary Economics*, *41*, 257–275.

- real business cycle model
- consumption habits
- investment adjustment costs
- compares return on several securities
- log–linearizes RBC model + log normal formulas for asset pricing

Firms

The representative firm maximizes its value:

$$\mathcal{E}_t \sum_{t+k}^{\infty} \beta^k \frac{\mu_{t+k}}{\mu_t} D_t$$

with

$$\begin{aligned}
Y_t &= A_t K_{t-1}^{\alpha} \left(X_t N_t\right)^{1-\alpha} \\
D_t &= Y_t - W_t Nt - I_t \\
K_t &= (1-\delta) K_{t-1} + \left(\frac{a_1}{1-\xi} \left(\frac{I_t}{K_{t-1}} \right)^{1-\frac{1}{x}} + a_2 \right) K_{t-1} \\
\log A_t &= \rho \log A_{t-1} + e_t \\
X_t &= (1+g) X_{t-1}
\end{aligned}$$

Households

The representative households maximizes current value of future utility:

$$\mathcal{E}_t \sum_{k=0}^{\infty} \beta^k \frac{(C_t - \chi C_{t-1})^{1-\tau}}{1-\tau}$$

subject to the following budget constraint:

$$W_t N_t + D_t = C_t$$

and with $N_t = 1$. Good market equilibrium imposes

$$Y_t = C_t + I_t$$

Interest rate

Risk free interest rate:

$$r_f = \frac{1}{\mathcal{E}_t \left\{ \beta g^{-\tau} \frac{\mu_{t+1}}{\mu_t} \right\}}$$

where μ_t is the utility of a marginal unit of consumption in period t.

$$\mu_t = (c_t - \chi c_{t-1}/g)^{-\tau} - \chi\beta\,(g c_{t+1} - \chi c_t)^{-\tau}$$

Rate of return

Rate of return of firms

$$
\begin{aligned}
r_t &= \mathcal{E}_t \Bigg\{ a_1 \left(\frac{gi_t}{k_{t-1}} \right)^{-\frac{1}{\xi}} \Bigg(\alpha z_{t+1} g^{1-\alpha} k_t^{\alpha-1} \\
&\quad + \frac{1 - \delta + \frac{a_1}{1-\frac{1}{\xi}} \left(\frac{gi_{t+1}}{k_t} \right)^{1-\frac{1}{\xi}} + a_2}{a_1 \left(\frac{gi_{t+1}}{k_t} \right)^{-\frac{1}{\xi}}} - \frac{gi_{t+1}}{k_t} \Bigg) \Bigg\}
\end{aligned}
$$

jermann98.mod

```
//----------------------------------------------------------------------
// 1. Variable declaration
//----------------------------------------------------------------------

var c, d, erp1, i, k, r1, rf1, w, y, z, mu;
varexo ez;
```

(continued)

```
  //----------------------------------------------------------------------
// 2. Parameter declaration and calibration
//----------------------------------------------------------------------

parameters alf, chihab, xi, delt, tau, g, rho, a1, a2, betstar, bet;

alf         = 0.36;     // capital share in production function
chihab      = 0.819;    // habit formation parameter
xi          = 1/4.3;    // capital adjustment cost parameter
delt        = 0.025;    // quarterly deprecition rate
g           = 1.005;    //quarterly growth rate (note zero growth =>g=1)
tau         = 5;        // curvature parameter with respect to c
rho         = 0.95;     // AR(1) parameter for technology shock

a1          = (g-1+delt)^(1/xi);
a2          = (g-1+delt)-(((g-1+delt)^(1/xi))/(1-(1/xi)))*
               ((g-1+delt)^(1-(1/xi)));
betstar     = g/1.011138;
bet         = betstar/(g^(1-tau));
```

(continued)

```
//----------------------------------------------------------------------
// 3. Model declaration
//----------------------------------------------------------------------

model;
g*k   = (1-delt)*k(-1)  +  ((a1/(1-1/xi))*(g*i/k(-1))^(1-1/xi)+a2)*k(-1);
d     = y - w - i;
w     = (1-alf)*y;
y     = z*g^(-alf)*k(-1)^alf;
c     = w + d;
mu    = (c-chihab*c(-1)/g)^(-tau)-chihab*bet*(c(+1)*g-chihab*c)^(-tau);
mu    = (betstar/g)*mu(+1)*(a1*(g*i/k(-1))^(-1/xi))*(alf*z(+1)*g^(1-alf)*
        (k^(alf-1))+((1-delt+(a1/(1-1/xi))*(g*i(+1)/k)^(1-1/xi)+a2))/
        (a1*(g*i(+1)/k)^(-1/xi))-g*i(+1)/k);
log(z) = rho*log(z(-1)) + ez;
```

(continued)

```
rf1  = 1/expectation(0)((betstar/g)*mu(+1)/mu);
r1   = (a1*(g*i/k(-1))^(-1/xi))*(alf*z(+1)*g^(1-alf)*(k^(alf-1))+
       (1-delt+(a1/(1-1/xi))*(g*i(+1)/k)^(1-1/xi)+a2)/
       (a1*(g*i(+1)/k)^(-1/xi))-g*i(+1)/k);
erp1 = r1 - rf1;

end;
```

(continued)

```
steady_state_model;
rf1     = (g/betstar);
r1      = (g/betstar);
erp1    = r1-rf1;
z       = 1;
k       = (((g/betstar)-(1-delt))/(alf*g^(1-alf)))^(1/(alf-1));
y       = (g^(1-alf))*k^alf;
w       = (1-alf)*y;
i       = (1-(1/g)*(1-delt))*k;
d       = y - w - i;
c       = w + d;
mu      = ((c-(chihab*c/g))^(-tau))-chihab*bet*((c*g-chihab*c)^(-tau));
ez      = 0;
end;
```

(continued)

```
steady;

shocks;
var ez; stderr 0.01;
end;

stoch_simul (order=2) rf1, r1, erp1, y, z, c, d, mu, k;
```

3rd order approximation

- same principle of derivation as 2nd order
- Don't forget option *periods=* in order to compute empirical moments